In[1]:=
```
(* We start by writing the ansatz *)
Celine[f_, n_, k_, r_, s_] :=
  Module[{ansatz},
   ansatz = Sum[c[i, j] * (f /. {n → n + i, k → k + j}), {i, 0, r}, {j, 0, s}]
  ];
Celine[Binomial[n, k], n, k, 1, 1]
```

Out[2]=
```
Binomial[n, k] c[0, 0] + Binomial[n, 1 + k] c[0, 1] +
 Binomial[1 + n, k] c[1, 0] + Binomial[1 + n, 1 + k] c[1, 1]
```

In[3]:=
```
(* We have to divide this ansatz by f and simplify. *)
Celine[f_, n_, k_, r_, s_] :=
  Module[{ansatz},
   ansatz =
    FunctionExpand[Sum[c[i, j] * (f /. {n → n + i, k → k + j}), {i, 0, r}, {j, 0, s}] / f]
  ];
Celine[Binomial[n, k], n, k, 1, 1]
```

Out[4]= $(-k+n)\left(\frac{c[0,0]}{-k+n}+\frac{c[0,1]}{1+k}+\frac{(1+n)\,c[1,0]}{(-1+k-n)\,(k-n)}+\frac{(1+n)\,c[1,1]}{(1+k)\,(-k+n)}\right)$

In[5]:=
```
(* Let's try a more complicated example *)
expr = Product[(Table[RandomInteger[{-10, 10}], {3}].{n, k, 1})!^((-1)^i), {i, 6}]
```

Out[5]= $\frac{(-2-9k-4n)!\,(-2+9k-3n)!\,(-3+3k+7n)!}{(8-k+3n)!\,(-8+k+5n)!\,(5+2k+9n)!}$

In[6]:=
```
(* This is quite slow *)
Timing[Celine[expr, n, k, 1, 1];]
```

Out[6]= {6.2635, Null}

In[7]:=
```
(* Better perform the simplification on each part separately *)
Celine[f_, n_, k_, r_, s_] :=
  Module[{ansatz},
   ansatz =
    Sum[c[i, j] * FunctionExpand[(f /. {n → n + i, k → k + j}) / f], {i, 0, r}, {j, 0, s}]
  ];
Celine[Binomial[n, k], n, k, 1, 1]
```

Out[8]= $c[0,0]+\frac{(-k+n)\,c[0,1]}{1+k}+\frac{(1+n)\,c[1,0]}{1-k+n}+\frac{(1+n)\,c[1,1]}{1+k}$

In[9]:=
```
(* Much better. *)
Timing[Celine[expr, n, k, 1, 1];]
```

Out[9]= {0.050475, Null}

In[10]:=
```
(* But what about this? :-( *)
Timing[Celine[expr, n, k, 10, 10];]
```

Out[10]= {90.9673, Null}

In[11]:=
```
(* Using the rational functions (certificates) of the hypergeometric input *)
Celine[f_, n_, k_, r_, s_] :=
  Module[{u, v, ansatz},
   u = FunctionExpand[(f /. n → n + 1) / f];
   v = FunctionExpand[(f /. k → k + 1) / f];
   ansatz = Table[Product[(u /. {n → n + i1, k → k + j}), {i1, 0, i - 1}] *
      Product[(v /. k → k + j1), {j1, 0, j - 1}], {i, 0, r}, {j, 0, s}]
  ];
```

In[12]:=
```
(* This is acceptable *)
Timing[Celine[expr, n, k, 10, 10];]
```

Out[12]= {0.185556, Null}

In[13]:=
```
(* Basic version of Sister Celine's algorithm *)
Celine[f_, n_, k_, r_, s_] :=
  Module[{u, v, mat, ns, rec},
   u = FunctionExpand[(f /. n → n + 1) / f];
   v = FunctionExpand[(f /. k → k + 1) / f];
   mat = Flatten[Table[Product[(u /. {n → n + i1, k → k + j}), {i1, 0, i - 1}] *
       Product[(v /. k → k + j1), {j1, 0, j - 1}], {i, 0, r}, {j, 0, s}]];
   (* Multiply with the common denominator *)
   mat = Together[mat * (PolynomialLCM @@ Denominator[mat])];
   (* Coefficient comparison w.r.t. k *)
   mat = Transpose[PadRight[CoefficientList[#, k] & /@ mat]];
   (* Compute the kernel *)
   ns = NullSpace[mat];
   If[ns === {}, Return[{}]];
   (* Each kernel vector gives a k-free recurrence. *)
   (* Sum these by replacing f[n+i,k+j] by SUM[n+i]. *)
   rec = ns.Flatten[Table[SUM[n + i], {i, 0, r}, {j, 0, s}]];
   (* Simplify the obtained recurrences and return them. *)
   rec = Collect[Numerator[Together[#]], SUM[_], Expand] & /@ rec;
   Return[rec];
  ];
```

In[14]:=
```
(* It works! *)
Celine[Binomial[n, k], n, k, 1, 1]
```

Out[14]= {-2 SUM[n] + SUM[1 + n]}

In[15]:=
```
Celine[Binomial[n, k]^2, n, k, 2, 2]
```

Out[15]=
```
{(-6 - 4 n) SUM[1 + n] + (2 + n) SUM[2 + n]}
```

In[16]:=
```
FullSimplify[%[[1]] /. SUM[n_] :> Binomial[2 n, n]]
```

Out[16]=
```
0
```

In[17]:=
```
(* Sister Celine's algorithm with
  (a quick-and-dirty version of) Verbaeten completion *)
Celine[f_, n_, k_, r_, s_, e_] :=
  Module[{u, v, mat, den, deg, idx, ns, rec},
   u = FunctionExpand[(f /. n → n + 1) / f];
   v = FunctionExpand[(f /. k → k + 1) / f];
   mat = Table[Product[(u /. {n → n + i1, k → k + j}), {i1, 0, i - 1}] *
      Product[(v /. k → k + j1), {j1, 0, j - 1}], {i, 0, r + e}, {j, 0, s + e}];
   den = PolynomialLCM @@ Denominator[Flatten[mat[[1 ;; r + 1, 1 ;; s + 1]]]];
   mat = Together[den * mat];
   deg = Max[Exponent[Flatten[mat[[1 ;; r + 1, 1 ;; s + 1]]], k]];
   mat = Flatten[mat];
   idx = Select[Range[Length[mat]],
     Denominator[mat[[#]]] === 1 && Exponent[mat[[#]], k] ≤ deg &];
   mat = mat[[idx]];
   mat = Transpose[PadRight[CoefficientList[#, k] & /@ mat]];
   ns = NullSpace[mat];
   If[ns === {}, Return[{}]];
   rec = ns.(Flatten[Table[SUM[n + i], {i, 0, r + e}, {j, 0, s + e}]][[idx]]);
   rec = Collect[Numerator[Together[#]], SUM[_], Expand] & /@ rec;
   Return[rec];
  ];
```

In[18]:=
```
(* Compare: original Celine's algorithm *)
Timing[Celine[(-1)^k * Binomial[2 n, n + k]^2, n, k, 2, 4]]
```

Out[18]=
```
{0.511511, {(112 + 400 n + 416 n^2 + 128 n^3) SUM[n] +
   (-110 - 288 n - 240 n^2 - 64 n^3) SUM[1 + n] + (18 + 45 n + 34 n^2 + 8 n^3) SUM[2 + n]}}
```

In[19]:=
```
(* and with Verbaeten completion *)
Timing[Celine[(-1)^k * Binomial[2 n, n + k]^2, n, k, 1, 3, 1]]
```

Out[19]=
```
{0.102244, {(112 + 400 n + 416 n^2 + 128 n^3) SUM[n] +
   (-110 - 288 n - 240 n^2 - 64 n^3) SUM[1 + n] + (18 + 45 n + 34 n^2 + 8 n^3) SUM[2 + n]}}
```

In[20]:=
```
(* Another example: Apery numbers *)
Timing[Celine[Binomial[n, k]^2 * Binomial[n + k, k]^2, n, k, 4, 3]]
Timing[Celine[Binomial[n, k]^2 * Binomial[n + k, k]^2, n, k, 2, 3, 2]]
```

Out[20]= $\{10.3229, \{(-504 - 2076\, n - 3408\, n^2 - 2832\, n^3 - 1248\, n^4 - 276\, n^5 - 24\, n^6)\, SUM[n] + (63\,000 + 194\,316\, n + 245\,760\, n^2 + 162\,672\, n^3 + 59\,256\, n^4 + 11\,232\, n^5 + 864\, n^6)\, SUM[1 + n] + (-277\,560 - 734\,604\, n - 798\,792\, n^2 - 457\,224\, n^3 - 145\,392\, n^4 - 24\,360\, n^5 - 1680\, n^6)\, SUM[2 + n] + (224\,280 + 564\,636\, n + 578\,136\, n^2 + 308\,280\, n^3 + 90\,360\, n^4 + 13\,824\, n^5 + 864\, n^6)\, SUM[3 + n] + (-9216 - 22\,272\, n - 21\,696\, n^2 - 10\,896\, n^3 - 2976\, n^4 - 420\, n^5 - 24\, n^6)\, SUM[4 + n]\}\}$

Out[21]= $\{4.05575, \{(-504 - 2076\, n - 3408\, n^2 - 2832\, n^3 - 1248\, n^4 - 276\, n^5 - 24\, n^6)\, SUM[n] + (63\,000 + 194\,316\, n + 245\,760\, n^2 + 162\,672\, n^3 + 59\,256\, n^4 + 11\,232\, n^5 + 864\, n^6)\, SUM[1 + n] + (-277\,560 - 734\,604\, n - 798\,792\, n^2 - 457\,224\, n^3 - 145\,392\, n^4 - 24\,360\, n^5 - 1680\, n^6)\, SUM[2 + n] + (224\,280 + 564\,636\, n + 578\,136\, n^2 + 308\,280\, n^3 + 90\,360\, n^4 + 13\,824\, n^5 + 864\, n^6)\, SUM[3 + n] + (-9216 - 22\,272\, n - 21\,696\, n^2 - 10\,896\, n^3 - 2976\, n^4 - 420\, n^5 - 24\, n^6)\, SUM[4 + n]\}\}$

In[22]:=
```
(* Note that we do not get the minimal second-order recurrence. *)
Arec = (1 + 3 n + 3 n^2 + n^3) SUM[n] +
  (-117 - 231 n - 153 n^2 - 34 n^3) SUM[1 + n] + (8 + 12 n + 6 n^2 + n^3) SUM[2 + n]
```

Out[22]= $(1 + 3\, n + 3\, n^2 + n^3)\, SUM[n] + (-117 - 231\, n - 153\, n^2 - 34\, n^3)\, SUM[1 + n] + (8 + 12\, n + 6\, n^2 + n^3)\, SUM[2 + n]$

In[23]:=
```
(* However, we can check that the larger
 recurrence is a consequence of the minimal one: *)
Together[%%[[2, 1]] //.
  SUM[n + i_ /; i ≥ 2] :> Solve[(Arec /. n → n + i - 2) == 0, SUM[n + i]][[1, 1, 2]]]
```

Out[23]= 0